

# Les sons préprogrammés de l'Oric (fin)

## 3e partie : ZAP et OUPS

par André C.

### Introduction

Ces deux derniers sons ont une structure différente de celle des 5 sons préprogrammés précédents. Ils n'utilisent pas d'enveloppe, mais font appel à des

boucles de temporisation pour contrôler l'évolution du son.

### LA COMMANDE ZAP

Voici la routine copiée et adaptée de la Rom :

```
org $9800 ; Adresse d'implantation sous-programme + data
PSG01P EQU $F590 ; Envoie le param X dans le registre A du PSG
PSG14P EQU $FA86 ; Envoie au PSG les 14 param situés à l'adresse XY
ZAP LDX #LOW PARAM ; LL, octet de poids faible de adr des 14 paramètres
LDY #HIGH PARAM ; HH, octet de poids fort de adr des 14 paramètres
JSR PSG14P
LLINIT LDA #$00 ; Octet de poids faible de la période initiale
TAX
BOUCL2 TXA ; Copie la période en cours dans le registre A
PHA ; Sauve cette période sur la pile (reste aussi dans X)
LDA #$00 ; A indexe le registre R0 du PSG
JSR PSG01P ; Envoie période X dans registre du PSG pointé par A
; Début de la temporisation
SEI
LDX #$00 ; 256 tours pour la boucle interne X
BOUCL1 DEX
BNE BOUCL1 ; Environ 1,28 ms pour la boucle interne X
CLI
; Fin de la temporisation
PLA ; Récupère la période en cours
TAX ; La copie dans X
INX ; Et l'incrémente
LLFIN CPX #$70 ; A-t-on atteint la période limite (octet faible)?
BNE BOUCL2 ; Non, on reboucle
LDA #$08 ; Oui, indexe registre 8 du PSG (volume du canal 1)
LDX #$00 ; Volume sonore nul
JMP PSG01P ; Envoie 0 dans le registre du PSG pointé par A
PARAM DB $00, $00, $00, $00, $00, $00, $00, $00 ; Paramètres R0 à R6
DB $3E, $0F, $00, $00, $00, $00, $00, $00 ; Paramètres R7 à R13
```

Voici ce que révèle le bloc des 14 paramètres :

- Contrôle des canaux en service : R7 = #3E, soit 0011 1110. Seul le canal 1 est activé.
- Période du canal 1 : R0/R1 = #0000. Soit une fréquence infinie que le PSG 9812 ne peut générer !
- Aucun bruit blanc n'est généré car R6 = #00.
- Volume du son du canal 1 : R8 = #0F (le maximum). Aucune enveloppe n'est mise en jeu.

Au premier abord, les paramètres de ZAP sont surprenants ! Le PSG est initialement mis en position de générer des ultra-sons, ce qu'il ne sait pas faire. Mais dans le code qui suit le chargement des 14 paramètres, une boucle est générée, comprenant une temporisation de 1,28 ms et une incrémentation de la période du 1<sup>er</sup> canal. Cette boucle tourne jusqu'à ce que la période atteigne #0070 ce qui correspond à 554 Hz : Le PSG génère

donc un son de plus en plus grave. Le processus se termine alors en envoyant #00 (volume sonore nul) dans le registre R8.

Si on augmente la temporisation par un facteur 10, afin de mieux percevoir ce qui sort du PSG au début de ZAP, on se rend compte que des criaileries atroces sont générées, comme si on était en train d'assassiner votre Oric !

Heureusement, avec la temporisation normale de ZAP, elles sont trop brèves pour être perçues. Mais il aurait été plus propre d'initialiser la période de départ du canal 1 avec une valeur un peu plus haute que #0000 !

### Variantes

La plus évidente est de réduire le volume sonore qui est actuellement au maximum. Un POKE # 982E suivi d'une valeur de #00 à #0F permet de l'ajuster finement.

Ceci mis à part, les autres possibilités de modifications sont infinies grâce au concept de temporisation permettant d'introduire une modification des paramètres du PSG9812. Rien n'empêche de compléter le profil du son produit. **ZAP est de loin le son préprogrammé le plus intéressant de l'Oric.**

Pour l'heure, nous serons plus modestes et fixerons une période de départ autre que #0000 et une période d'arrivée autre que #0070. Nous pourrions aussi modifier la temporisation, c'est-à-dire la durée de chaque fréquence générée, et par suite la durée totale du son produit.

Comme précédemment, la procédure consistera à charger ZAPLM en Ram et à POKer les valeurs à modifier. Cette routine est localisée de #9800 à #9833 et les adresses à POKer sont les suivantes :

- En #982E : Paramètre R8. Le volume du son du canal 1.
- En #9808 : Octet de poids faible de la période de départ (limitée à #FF, car un seul octet est pris en compte).
- En #981C : Octet de poids faible de la période finale (idem un seul octet pris en compte).
- En #9813 : Le nombre de tours de la boucle de

```

org $9800          ; Adresse d'implantation sous-programme + data
PSG01P EQU $F590   ; Envoie le param X dans le registre A du PSG
PSG14P EQU $FA86   ; Envoie au PSG les 14 param situés à l'adresse XY
OUPS   LDX #LOW PARAM ; LL de l'adresse des 14 paramètres
        LDY #HIGH PARAM ; HH de l'adresse des 14 paramètres
        JSR PSG14P
; Début de la temporisation
        SEI
YTOURS LDY #$60     ; Temporisation boucle externe (96 tours)
        LDX #$00     ; Temporisation boucle interne (256 tours)
BCOUL1 DEX
        BNE BCOUL1   ; Délai boucle interne de 1,28 ms
        DEY
        BNE BCOUL1   ; Délai total de 0,12s (boucles X et Y)
        CLI
; Fin de la temporisation
        LDA #$07     ; Registre 7 du PSG : contrôle des canaux
        LDX #$3F     ; 0011 1111 tous inactivés
        JMP PSG01P
PARAM  DB $46, $00, $00, $00, $00, $00, $00, $00 ; Paramètres R0 à R6
        DB $3E, $0F, $00, $00, $00, $00, $00 ; Paramètres R7 à R13

```

Voici ce que révèle le bloc des 14 paramètres :

- Contrôle des canaux en service : R7 = #3E, soit 0011 1110. Seul le canal 1 est activé.
- Période du canal 1 : R0/R1 = #0046 (environ un LA de l'octave 4).
- Aucun bruit blanc n'est généré car R6 = #00.
- Volume du son du canal 1 : R8 = #0F (le maximum). Aucune enveloppe n'est mise en jeu.

Le PSG 9812 génère un LA de l'octave 4 avec un volume sonore maximum suivi d'une temporisa-

temporisation. Dans l'état actuel de la routine (une boucle de 256 répétitions), il est seulement possible de réduire cette durée. Pour l'augmenter, il faudra mettre en place une deuxième boucle.

Le programme ZAP1.BAS offre quelques échantillons tout prêts tandis que ZAP2.BAS permet d'expérimenter soi-même de nouveaux sons, en ajustant le volume sonore, la temporisation, les périodes de départ et de fin.

Quelques remarques :

1. La durée des sons produits est très variable compte tenu de la gamme des fréquences à parcourir.
2. Les sons produits vont de l'aigu vers le grave. Or les sons aigus semblent moins audibles que les graves (à moins que ce ne soit un problème personnel). Il s'en suit que toutes les variantes se ressemblent car elles finissent par des sons graves mieux perçus.

#### LA COMMANDE OUPS

Voici le code d'OUPS transposé et adapté pour l'Atmos à partir du code original d'OUPS pour Tel-estrat :

tion de 0,12 s avant que finalement le canal 1 soit inactivé. Simple non ?

#### Variantes

OUPS repose sur 3 paramètres : Période et volume du son, durée de la temporisation. La période du canal 1 (R0/R1) peut être comprise entre #0000 et #0FFF. Le volume sonore (R8) est ajustable de #00 à #0F. La durée du son peut varier de 1,28 ms (avec Y=1) et 0,33 s (avec Y=#FF). On peut évidemment générer une infinité de sons en combinant ces 3 paramètres.

Après avoir chargé OUPSLM en Ram, il suffit de poker les valeurs à modifier. Compte tenu de l'adresse d'implantation d'OUPS en Ram (de #9800 à #9827), les adresses à poker sont :

- En #981A-#981B : Période du canal 1 (R0/R1).
- En #9809 : Nombre de tours de la boucle externe Y.
- En #9822 : Volume sonore du canal 1 (R8)

Comme pour les autres sons, je vous propose deux petits programmes Basic : Le premier OUPS1.BAS propose divers échantillons de variantes d'OUPS et le second OUPS2.BAS permet d'expérimenter soi-même toutes les combinaisons possibles des 3 paramètres d'OUPS.

#### Le programme OUPS1.BAS

Le menu propose 20 périodes correspondant à la fréquence des notes de DO de l'octave 1 au DO de l'octave 6, puis du LA de l'octave 1 au LA de l'octave 6 et enfin toute la gamme de l'octave 4. Les deux autres paramètres (durée et volume du son) n'ont pas été modifiés et sont ceux du OUPS d'origine.

#### Le programme OUPS2.BAS

Le menu vous propose de tester vous-même tous les sons de votre choix. On peut fixer indépendamment période, volume et durée et écouter ce que cela donne.

#### LA COMMANDE ZAAP

Voici une nouvelle famille de sons, les "ZAAP", basés sur la routine ZAP, mais dont la durée serait augmentée grâce à une boucle supplémentaire. Non seulement cela permettra d'obtenir des sons plus longs, mais aussi de "normaliser" leur durée, qui varie selon le nombre de périodes balayées. Au final, avec cette "normalisation", des sons de durée similaire pourront plus facilement être comparés.

#### Listing modifié :

Le nouveau listing ZAAPLM.ASM peut être obtenu à partir du listing ZAPLM.ASM en remplaçant la section située entre les lignes ";Début de la temporisation" et ";Fin de la temporisation" par la même section prise dans OUPSLM.ASM. Cette boucle supplémentaire permet des temporisations de 1,23 ms (pour Y=#01) à 315 ms (pour Y=#00, c'est-à-dire #100 en réalité). La temporisation comporte une partie fixe (boucle interne de 1,28 ms environ) et une partie ajustable, la 2<sup>e</sup> boucle ou boucle externe, qui joue un rôle multiplicateur. La durée totale théorique en ms d'un ZAAP est égale à 1,28 x(période finale – période initiale) x(nombre de tours de 2<sup>e</sup> boucle). La brièveté des sons ne permet pas de vérifier les valeurs réelles obtenues.

Le programme ZAAP1.BAS montre ce que ça donne avec les octaves 3 à 7 lorsque la temporisation a été ajustée pour obtenir des sons de durée totale de 160, 320, 640 et 1300 ms.

Voici les périodes de départ et de fin utilisées pour chaque octave :

#### Résultats

On observe une nette progressivité dans les résultats obtenus avec les octaves 7 à 3 et pour chacune avec les durées croissantes. C'est l'octave 7 (la plus aiguë) qui est la moins satisfaisante : La durée 640 ms et surtout la durée 1300 ms révèlent des distorsions peu agréables. Mais dans l'ensemble, cela représente une belle bibliothèque de sons cohérents dérivés de ZAP.

Le programme ZAAP2.BAS vous permettra de procéder à vos propres tests en modifiant, les périodes de départ et de fin, ainsi que le nombre de boucles Y et le volume sonore. Le tableau ci-dessous vous sera probablement de quelque aide.

Octave	3	4	5	6	7
Période de départ	#7F	#3F	#1F	#0F	#07
Période de fin	#FE	#7E	#3E	#1E	#0E
Et voici les valeurs de Y utilisées pour avoir une durée totale "normalisée" :					
Durée totale env. 160 ms, Y=	18	8	4	2	1
Durée totale env. 320 ms, Y=	36	17	8	4	2
Durée totale env. 640 ms, Y=	73	34	16	8	4
Durée totale env. 1300 ms, Y=	146	68	33	16	8

Vous trouverez tous les programmes produits au cours de ce travail dans le fichier SonsPreprogr3.zip qui accompagne cet article.

